DECUS
PROGRAM LIBRARY

DECUS NO.	8-268
TITLE	MINILOADER AND MINILOADER PUNCH
AUTHOR	G. J. Flanagan
COMPANY	The University of Newcastle Upon Tyne Newcastle Upon Tyne, England
DATE	March 3, 1970
SOURCE LANGUAGE	PAL III

MINILOADER AND MINILOADER PUNCH

DECUS Program Library Write-up

DECUS No. 8-268

THE MINILOADER

THIS IS A PROGRAM WHICH IS DESIGNED TO BE LOADED AS EASILY AS POSSIBLE FROM THE SWITCH REGISTER INTO A PDP8 COMPUTER. ONCE LOADED, THE PROGRAM IS CAPABLE OF LOADING AND STARTING ANY OTHER PROGRAM PUNCHED IN THE CORRECT FORM, THOUGH BECAUSE OF THE LACK OF ANY CHECKING FACILITY IT WOULD NORMALLY LOAD A STANDARD BINARY LOADER.

THERE ARE THREE VERSIONS OF THE PROGRAM, CHOSEN ACCORDING TO THE EQUIPMENT INSTALLED. THE LONGEST NEEDS ELEVEN AND THE SHORTEST ONLY NINE SETTINGS OF THE SWITCH REGISTER INCLUDING THE "START LOADING" AND "START" ADDRESSES. FROM THE TAPE THE LOADER WILL COMPLETE LOADING ITSELF, LOAD A PROGRAM, AND EITHER RESTORE ITSELF TO A STATE IN WHICH IT CAN START AGAIN, OR START THE PROGRAM IT HAS LOADED. ANY REGISTERS MAY BE OMITTED FROM A SEQUENCE BEING LOADED.

A TAPE PUNCHED IN MINILOADER FORMAT WILL WORK WITH ANY OF THE THREE VERSIONS OF THE MINILOADER. IT CAN BE READ EITHER BY THE TELETYPE READER OR THE HIGH SPEED READER, DEPENDENT ON WHICH VERSION IS USED. THE MINILOADER FORMAT IS SUCH THAT, IF THE PROGRAM IS TO BE LOADED AT THE HIGH-NUMBERED END OF THE FIELD, THE LENGTH OF THE TAPE IS JUST OVER HALF THE LENGTH OF THE SAME PROGRAM PUNCHED IN RIM FORMAT.

THE MINILOADER

ADDRESS	TELETYPE READER	HIGH SPEED READER WITH BUFFER	HIGH SPEED READER WITHOUT BUFFER	
	A	B	C	
7762	7106			
7763	7106	6014		
7764	**7106**	**6011**	**6016**	EXIT HERE
7765	3354	5364	6011	
7766	6031	6012	5365	
7767	5366	7430	7430	
7770	6036	3776	3776	
7771	1354	7106	7106	
7772	7430	7106	7106	
7773	3776	7106	7106	
7774	5362	5363	5364	CHANGED TO 7000 BY A TAPE WHICH DOES
7775	5362	5363	5364	NOT LOAD AN EXIT ORDER
7776	7774	7774	7774	

7774 IS ALSO THE START ADDRESS.

VERSION B WILL WORK WITHOUT A READER BUFFER, BUT IF VERSION C WERE USED WHERE THERE IS A BUFFER IT WOULD ALLOW AN UNWANTED CHARACTER LEFT IN THE BUFFER TO BE LOADED. VERSION B IS THEREFORE USED ON TAPES, BUT VERSION C IS LOADED BY HAND IF IT CAN BE USED, BECAUSE IT IS EASIER TO LOAD.

AS LOADED, THE PROGRAM CAN ONLY LOAD ADDRESS 7774. THE FIRST TWO CHARACTERS FROM THE PAPER TAPE COMPLETE THE LOADING OF THE PROGRAM BY PUTTING "ISZ 7776" IN THIS REGISTER. DOING THIS SAVES SETTING ONE NUMBER IN THE SWITCH REGISTER AND ALSO ALLOWS THE PROGRAM TO ACCEPT AN INDEFINITE LENGTH OF LEADER TAPE BEFORE THE FIRST INSTRUCTION OCCURS. THE LEADER IS BLANK TAPE, WHICH DOES NOT SET ANY REGISTER TO ZERO BECAUSE IT KEEPS THE LINK AT ZERO, CAUSING THE DCA INSTRUCTION TO BE SKIPPED. ONCE THE FIRST INSTRUCTION HAS BEEN READ, THE PROGRAM INCREMENTS THE ADDRESS AS OFTEN AS IT READS A CHARACTER. THIS MEANS THAT

IT CANNOT LOAD CONSECUTIVE ADDRESSES IN ONE PASS. DURING LOADING, THE EVEN-NUMBERED REGISTERS ARE LOADED FIRST, AND THEN THE ODD-NUMBERED REGISTERS. THE MINILOADER AVOIDS LOADING A REGISTER BY READING BLANK TAPE AT THE APPROPRIATE PLACE.

THE SECOND ADDRESS WHICH CAN BE LOADED IS 7776, WHICH STORES THE ADDRESS FOR THE LOADING INSTRUCTION. THE SECOND PAIR OF CHARACTERS ON THE TAPE SETS THIS TO TWO LESS THAN THE FIRST EVEN ADDRESS OF THE PROGRAM TO BE LOADED. WHEN ALL THE EVEN ADDRESSES TO 7760 HAVE BEEN LOADED, BLANK TAPE INCREMENTS THE ADDRESS REGISTER UNTIL 7776 CAN BE LOADED AGAIN. THE NEXT PAIR OF CHARACTERS SET IT TO TWO LESS THAN THE FIRST ODD ADDRESS. WHEN ALL THE ODD ADDRESSES TO 7761 ARE LOADED, THE TERMINATING INSTRUCTION IS LOADED AT ITS APPROPRIATE ADDRESS.

IF THE PROGRAM JUST LOADED IS NOT TO BE STARTED AUTOMATICALLY, THE TAPE RESTORES THE MINILOADER TO A STARTING CONDITION. IT IS THEN UNAFFECTED BY BLANK TRAILER TAPE. BECAUSE THE JUMP WHICH IS OVERWRITTEN BY THE FIRST ORDER ON A MINILOADER TAPE IS DIFFERENT IN EACH VERSION, IT IS REPLACED BY 7000 (NOP) BY THE LAST ORDER ON THE TAPE. THE APPROPRIATE JUMP IN THE NEXT REGISTER IS THEN EFFECTIVE TO ASSEMBLE THE FIRST ORDER ON THE NEXT TAPE. LOADING 7774 ALSO LEAVES 7776 SET TO 7774 REGARDLESS OF THE LENGTH OF ANY BLANK TRAILER TAPE WHICH MIGHT BE READ. THE TAPE IS STOPPED BY THE TAPE-OUT SWITCH AND THE PROGRAM MUST BE STOPPED BY THE CONSOLE SWITCH.

IF THE PROGRAM ON TAPE IS TO BE STARTED WHEN IT HAS BEEN LOADED, A SUITABLE JUMP INSTRUCTION IS LOADED AT 7764. THIS ALLOWS "RFC" TO BE PERFORMED WHEN VERSION B IS IN USE AND AVOIDS "RRB RFC" IF VERSION C IS USED. VERSION A HAS BEEN ARRANGED SO THAT THIS POSITION IS ALSO SUITABLE THERE.

ALL THE OPTIONS ARE SELECTED WHEN THE TAPE IS PUNCHED, AND THE APPROPRAIATE FORMAT IS PRODUCED BY THE "MINILOADER PUNCH" PROGRAM.

MINILOADER PUNCH

THIS PROGRAM PUNCHES PAPER TAPE IN A FORMAT WHICH ENABLES THE MINILOADER TO RELOAD A PROGRAM (REFERRED TO AS THE "SUBJECT PROGRAM") WHICH IS HELD IN THE CORE STORE.

REGISTER 20 (OCTAL) CONTAINS THE EXIT INSTRUCTION FOR THE MINILOADER. AS LOADED IT IS "JMP NEXT" TO START THE BINARY LOADER 48. IT MAY BE REPLACED BY ANOTHER JUMP INSTRUCTION OR SET TO ZERO. IF THIS REGISTER IS SET TO ZERO, A TAPE WILL BE PUNCHED WHICH RESTORES ANY VERSION OF THE MINILOADER TO A STARTING CONDITION WHEN LOADING IS COMPLETED.

THE LOWEST-NUMBERED ADDRESS OF THE SUBJECT PROGRAM IS SET IN THE SR, AND THE PUNCH PROGRAM STORES THE APPROPRIATE ODD AND EVEN ADDRESSES FOR THE MINILOADER. THE HIGHEST-NUMBERED ADDRESS IS ALWAYS 7761, THE REMAINDER BEING RESERVED FOR THE MINILOADER. BEFORE PUNCHING BEGINS, THE PUNCH PROGRAM HALTS FOR SWITCH OPTIONS WHICH ALLOW ANY REGISTER BETWEEN THESE LIMITS TO BE LEFT UNALTERED WHEN THE TAPE IS USED.

THE HIGH SPEED READER IS USED IF IT IS ON, AND THE HIGH SPEED PUNCH IS USED IF IT IS ON. OTHERWISE THE TELETYPE DEVICES ARE USED.

PRODUCT NAME: MINILOADER

DATE: NOVEMBER 1969

1. ABSTRACT

THE MINILOADER IS A ROUTINE FOR READING AND STORING INFORMATION CONTAINED IN MINILOADER CODED TAPES. THE NUMBER OF SETTINGS OF THE SWITCH REGISTER (SR) NEEDED TO LOAD IT IS MINIMISED. VERSIONS ARE GIVEN HERE FOR USE WITH THE ASR-33 PERFORATED TAPE READER AND WITH THE HIGH SPEED PHOTO-ELECTRIC READER.

2. REQUIREMENTS

2.1 EQUIPMENT

4K PDP-8 WITH ASR-33 OR PHOTO-ELECTRIC READER.

2.2 STORAGE

THIRTEEN REGISTERS (13 DECIMAL, 15 OCTAL).

3. USAGE

3.1 LOADING

TO PLACE THE MINILOADER INTO MEMORY VIA THE CONSOLE SWITCHES, PROCEED AS FOLLOWS:

3.1.1 SET THE FIRST ADDRESS (SEE LISTS FOR APPROPRIATE VERSION) IN THE SWITCH REGISTER (SR).

3.1.2 PRESS 'LOAD ADDRESS'.

3.1.3 SET THE FIRST INSTRUCTION IN THE SR.

3.1.4 PRESS 'DEPOSIT'.

3.1.5 IF THE NEXT NUMBER IS THE SAME AS THE LAST, REPEAT STEP 3.1.4. IF NOT, SET THE NEXT INSTRUCTION IN THE SR.

3.1.6 REPEAT FROM STEP 3.1.4 UNTIL ALL THE INSTRUCTIONS HAVE BEEN DEPOSITED.

3.1.7 PRESS 'LOAD ADDRESS' (THE STARTING ADDRESS BEING THE SAME AS THE LAST NUMBER LOADED).

3.2 CALLING SEQUENCE

NONE. THIS PROGRAM CANNOT BE CALLED AS A SUBROUTINE.

3.3 START-UP/ENTRY

3.3.1 PLACE THE PERFORATED TAPE, WHICH MUST BE IN MINILOADER FORMAT, IN THE APPROPRIATE READER.

3.3.2 IF THE ASR-33 IS BEING USED, SEE THAT IT IS ON LINE. IF THE PHOTO-ELECTRIC READER IS BEING USED, SEE THAT IT IS ON.

3.3.3 SEE THAT THE START ADDRESS (7774) IS IN THE PROGRAM COUNTER. IF NOT, SET THIS NUMBER IN THE SR AND PRESS 'LOAD ADDRESS'.

3.3.4 PRESS THE CONSOLE 'START' SWITCH.

3.3.5 IF THE ASR-33 IS BEING USED, MOVE THE READER CONTROL SWITCH TO 'START'.

3.4 ERRORS

THERE ARE NO ERROR STOPS IN THIS ROUTINE.

4. DESCRIPTION

4.1 THE THREE VERSIONS OF THIS ROUTINE ARE BASIC LOADERS WHICH ARE EACH DESIGNED FOR MAXIMUM EASE AND SPEED OF LOADING VIA THE SR. A TAPE MADE FOR ONE VERSION WILL WORK WITH THE OTHER VERSIONS. THE LOADER IS COMPLETED BY THE FIRST TWO CHARACTERS ON THE TAPE. THIS ALLOWS THE LOADER TO ACCEPT AN INDEFINITE LENGTH OF LEADER. THE LAST TWO CHARACTERS OF THE MINILOADER FORMAT EITHER RESTORE THE MINILOADER TO A STARTING CONDITION (NOT IDENTICAL WITH THAT LOADED VIA THE SR), OR START THE PROGRAM WHICH THE MINILOADER HAS LOADED. (THIS CHOICE IS MADE WHEN THE TAPE IS PUNCHED.)

4.2 BECAUSE THE MINILOADER DOES NO CHECKING, IT IS RECOMMENDED THAT IT IS USED ONLY TO LOAD A BINARY LOADER. THE MINILOADER CAN THEN START THE BINARY LOADER AND FURTHER PROGRAM IN BINARY FORMAT CAN BE LOADED FROM THE SAME TAPE. (SEE 'LOADING SYSTEM'.)

5. FORMAT

5.1 EXTERNAL DATA

5.1.1 A TAPE TO BE READ BY THIS PROGRAM MUST BE IN MINILOADER FORMAT. SUCH A TAPE CAN SET EVERY REGISTER IN THE RANGE STARTING AT THE REGISTER WHICH IS SPECIFIED WHEN THE TAPE IS PUNCHED AND ENDING AT REGISTER NUMBER 7761.

5.1.1.1 AN OPTION IN THE 'MINILOADER PUNCH' PROGRAM ALLOWS A TAPE TO BE PUNCHED WHICH WILL AVOID SETTING ANY SPECIFIED REGISTERS IN THE RANGE MENTIONED IN 5.1.1.

5.1.2 LEADER MUST BE BLANK TAPE. (ONE TO TWO FEET IS SUGGESTED.

5.1.2.1 LEADER IS INCLUDED IN THE FORMAT PUNCHED BY THE 'MINILOADER PUNCH' PROGRAM.

5.1.3 EACH TWELVE-BIT NUMBER TO BE LOADED IS REPRESENTED ON THE TAPE BY TWO CHARACTERS. THE FIRST HAS CHANNEL 8 NOT PUNCHED AND CHANNEL 7 PUNCHED, AND THE REMAINING CHANNELS REPRESENT THE SIX MOST SIGNIFICANT BITS OF THE NUMBER, CHANNEL 6 REPRESENTING THE MOST SIGNIFICANT BIT. THE SECOND CHARACTER HAS CHANNELS 8 AND 7 NOT PUNCHED, AND THE REMAINING CHANNELS REPRESENT THE REST OF THE NUMBER, CHANNEL 1 REPRESENTING THE LEAST SIGNIFICANT BIT.

5.1.3.1 THE NUMBERS OF ITEM 5.1.3 REPRESENT THE PROGRAM TO BE LOADED, REFERRED TO AS THE 'SUBJECT PROGRAM', AND CERTAIN ORDERS CONCERNED WITH THE OPERATION OF THE MINILOADER, ALL IN MACHINE LANGUAGE.

5.1.3.2 EACH PAIR OF THE CHARACTERS OF ITEM 5.1.3 IS ARRANGED IN A PATTERN AS FOLLOWS:

CHARACTER NUMBERS		CONTENT
1,	2	ISZ A (A=7776). THIS IS LOADED AT 7774 AND COMPLETES THE MINILOADER.
3,	4	TWO LESS THAN THE FIRST EVEN ADDRESS OF THE SUBJECT PROGRAM.
5,	6	THE CONTENTS OF THE FIRST EVEN ADDRESS OF THE SUBJECT PROGRAM.
7 TO	N	THE CONTENTS OF EACH SUCCESSIVE EVEN ADDRESS TO 7760.
N+1 TO	N+12	TWELVE BLANK CHARACTERS.
N+13,	N+14	TWO LESS THAN THE FIRST ODD ADDRESS OF THE SUBJECT PROGRAM.
N+15,	N+16	THE CONTENTS OF THE FIRST ODD ADDRESS OF THE SUBJECT PROGRAM.
N+17 TO	M	THE CONTENTS OF EACH SUCCESSIVE ODD ADDRESS TO 7761.
EITHER:		
M+1 TO	M+11	NINE BLANK CHARACTERS FOLLOWED BY 7000 AND THEN BLANK TRAILER,
OR:		
M+1 TO	M+3	ONE BLANK CHARACTER FOLLOWED BY THE JUMP INSTRUCTION WHICH TRANSFERS PROGRAM CONTROL TO THE SUBJECT PROGRAM.

5.1.3.3 THIS PATTERN IS PRODUCED BY THE 'MINILOADER PUNCH' PROGRAM.

6. PROGRAM

6.1 VERSION A, FOR ASR-33 TELETYPE READER.

6.1.1 PROGRAM LIST

```
ABS.    OCTAL   TAG      INSTRUCTION      COMMENTS
ADDR.   CONTENTS

7762    7106    B,       CLL RTL / CLEAR LINK AND ROTATE 6 LEFT
7763    7106             CLL RTL
7764    7106             CLL RTL / EXIT HERE WHEN LOADING IS COMPLETE
7765    3377             DCA WORD/ CLEAR 'WORD' OR STORE THE MOST
7766    6031             KSF     /       SIGNIFICANT HALF OF THE WORD
7767    5366             JMP .-1 / WAITING FOR CHARACTER
7770    6035             KRB     / READ BUFFER
7771    1377             TAD WORD/ TAD ZERO OR THE MOST SIGNIFICANT HALF
7772    7430             SZL     / SKIP UNLESS THE WORD IS ASSEMBLED
7773    3776             DCA I A / STORE WORD IF ASSEMBLED
7774    5362    START,   JMP B   / CHANGED TO 'NOP' BY A TAPE WHICH DOES
7775    5362             JMP B   /       NOT LOAD AN EXIT ORDER
7776    7774    A,       START   / ADDRESS TO BE LOADED
```

6.2 VERSION B, FOR HIGH SPEED PHOTO-ELECTRIC READER WITH READER BUFFER.

6.2.1 PROGRAM LIST

```
ABS.    OCTAL   TAG      INSTRUCTION      COMMENTS
ADDR.   CONTENTS
7763    6014    B,       RFC
7764    6011             RSF     / EXIT HERE WHEN LOADING IS COMPLETE
7765    5364             JMP .-1 / WAITING FOR CHARACTER
7766    6012             RRB
7767    7430             SZL     / SKIP UNLESS THE WORD IS ASSEMBLED
7770    3776             DCA I A / STORE WORD IF ASSEMBLED
7771    7106             CLL RTL / CLEAR LINK AND ROTATE 6 LEFT
7772    7106             CLL RTL
7773    7106             CLL RTL
7774    5363    START,   JMP B   / CHANGED TO 'NOP' BY A TAPE WHICH DOES
7775    5363             JMP B   /       NOT LOAD AN EXIT ORDER
7776    7774    A,       START   / ADDRESS TO BE LOADED
```

6.3 VERSION C, FOR HIGH SPEED PHOTO-ELECTRIC READER WITHOUT READER BUFFER.

6.3.1 PROGRAM LIST

```
ABS.    OCTAL   TAG      INSTRUCTION      COMMENTS
ADDR.   CONTENTS

7764    6016    B,       RRB RFC / EXIT HERE WHEN LOADING IS COMPLETE
7765    6011             RSF
7766    5365             JMP .-1 / WAITING FOR CHARACTER
7767    7430             SZL     / SKIP UNLESS THE WORD IS ASSEMBLED
7770    3776             DCA I A / STORE WORD IF ASSEMBLED
7771    7106             CLL RTL / CLEAR LINK AND ROTATE 6 LEFT
7772    7106             CLL RTL
7773    7106             CLL RTL
7774    5364    START,   JMP B   / CHANGED TO 'NOP' BY A TAPE WHICH DOES
7775    5364             JMP B   /       NOT LOAD AN EXIT ORDER
7776    7774    A,       START   / ADDRESS TO BE LOADED
```

6.4 VERSION B WILL WORK WITH OR WITHOUT A READER BUFFER AND IS THEREFORE USED ON TAPES. VERSION C IS SATISFACTORY ONLY WHERE THERE IS NO READER BUFFER. IT IS USED WHERE POSSIBLE WHEN LOADED FROM THE SR, BECAUSE IT IS EASIER TO LOAD.

7. USING THE MINILOADER WITH AN EXTENDED MEMORY

7.1 THE MINILOADER MAY RUN IN ANY MEMORY FIELD, PROVIDED THAT THE DATA FIELD REGISTER AND THE INSTRUCTION FIELD REGISTER ARE BOTH SET TO N (A NUMBER FROM 0 TO 7) WHERE N IS THE NUMBER OF THE MEMORY FIELD IN WHICH THE MINILOADER IS TO BE PLACED (SEE 'LOADING SYSTEM'). THIS IS DONE AS FOLLOWS:

7.2 SET THE DATA FIELD EXTENSION OF THE SWITCH REGISTER TO N.

7.3 SET THE INSTRUCTION FIELD EXTENSION OF THE SWITCH REGISTER TO N.

7.4 FOLLOW THE PROCEEDURE IN STEPS 3.1.1 TO 3.1.7 ABOVE.

PRODUCT NAME: MINILOADER PUNCH

DATE: NOVEMBER 1969

1. ABSTRACT

THE MINILOADER PUNCH PROGRAM TRANSFERS INFORMATION CONTAINED IN SELECTED REGISTERS OF CORE MEMORY TO PUNCHED PAPER TAPE IN MINILOADER FORMAT. THE INFORMATION WOULD NORMALLY BE A BINARY LOADER PROGRAM. THE HIGH SPEED PUNCH IS USED IF IT IS AVAILABLE AND ON, OTHERWISE THE ASR-33 PUNCH IS USED. THE PUNCH PROGRAM OCCUPIES LOW-NUMBERED REGISTERS BECAUSE IT IS INTENDED TO RECORD PROGRAM RESIDING IN HIGH-NUMBERED REGISTERS.

2. REQUIREMENTS

2.1 EQUIPMENT

PDP-8 WITH ASSOCIATED ASR-33 OR HIGH SPEED PUNCH.

2.2 STORAGE

116 DECIMAL (164 OCTAL).

3. LOADING OR CALLING PROCEEDURE

3.1 THIS PROGRAM IS LOADED USING A BINARY LOADER.

3.2 CALLING SEQUENCE

NONE. THIS PROGRAM CANNOT BE CALLED AS A SUBROUTINE.

4. USING THE PROGRAM

4.1 SWITCH SETTINGS

THE SWITCH REGISTER IS USED TO SELECT CERTAIN OPTIONS, AND TO ENTER THE LOWEST REGISTER NUMBER TO BE RECORDED. IF ANY REGISTER BETWEEN THIS AND 7761 MUST BE LEFT UNALTERED WHEN THE RESULTING TAPE IS LOADED, THAT REGISTER MUST BE SET TO ZERO BEFORE PUNCHING BEGINS.

4.2 THE PROGRAM IS NORMALLY USED TO PUNCH A TAPE WHICH, WHEN LOADED, WILL TRANSFER CONTROL TO THE PROGRAM READ FROM THE TAPE. IF THIS FACILITY IS TO BE USED, LOCATION 20 (OCTAL) MUST FIRST BE LOADED WITH A JUMP INSTRUCTION TO A START ADDRESS FOR THE PROGRAM TO BE PUNCHED. THIS ADDRESS MUST BE IN THE RANGE 7600 TO 7761 (OCTAL) (OR IN PAGE 0). IF A

TAPE IS REQUIRED WHICH DOES NOT START THE PROGRAM IT LOADS, REGISTER 20 MUST BE SET TO ZERO, AS IT IS WHEN THIS PROGRAM IS LOADED.

4.3 START UP / ENTRY

IF LOADED BY A LOADER FOR SELF-STARTING PROGRAM TAPES, 20 WILL BE IN THE PC. OTHERWISE SET 20 IN THE SR, AND PRESS 'LOAD ADDRESS'.

4.3.1 WHEN REGISTER 20 HAS BEEN EXAMINED OR LOADED FROM THE SR, THE START ADDRESS (21) IS IN THE PROGRAM COUNTER.

4.3.2 SET THE LOWEST-NUMBERED ADDRESS TO BE RECORDED IN THE SR.

4.3.3 PRESS 'START'. THE PROGRAM HALTS AT 33, WITH THE ADDRESS IN AC.

4.3.4 TO PUNCH A TAPE WHICH WILL SET ALL THE REGISTERS FROM THAT SPECIFIED IN STEP 4.3.2 TO 7761, SEE THAT SR0 = 1 AND SR1 = 1.

4.3.5 SEE THAT THE PUNCH IS ON. IF THE ASR-33 IS USED, SEE THAT IT IS ON LINE.

THE HIGH SPEED PUNCH IS USED IF IT IS AVAILABLE AND ON. OTHERWISE THE ASR-33 PUNCH IS USED.

4.3.6 PRESS 'CONTINUE'.

4.3.7 IF ANY REGISTER IN THE ABOVE SEQUENCE IS TO BE LEFT UNALTERED WHEN THE RESULTING TAPE IS USED, THAT REGISTER WILL HAVE BEEN SET TO ZERO BEFORE PUNCHING STARTS (SEE 4.1 ABOVE).

IF SR0 IS AT ZERO WHEN THE PUNCHING FOR ANY SUCH REGISTER TAKES PLACE, THAT REGISTER WILL BE UNALTERED BY THE RESULTING TAPE.

IF SR0 = 1 AT THIS TIME, THAT REGISTER WILL BE SET TO ZERO BY THE RESULTING TAPE.

IF SR1 = 0, THE PROGRAM WILL HALT BEFORE EACH SUCH PUNCHING SO THAT SR0 MAY BE RESET.

TO FACILITATE RECOGNITION OF THE REGISTER CONCERNED, ITS ADDRESS IS IN THE AC WHEN THE PROGRAM HALTS.

NOTE THAT ALL THE EVEN-NUMBERED ADDRESSES ARE RECORDED FIRST AND THEN ALL THE ODD-NUMBERED ADDRESSES.

4.3.8 WHEN PUNCHING IS COMPLETE THE PROGRAM HALTS WITH 20 IN THE PC AGAIN. THE SEQUENCE MAY BE REPEATED FROM 4.2.

5. DETAILS OF OPERATION AND STORAGE

AFTER ENTRY, THE LOWEST-NUMBERED ADDRESS IS READ FROM THE SR AND IF IT IS EVEN IT IS INCREMENTED. 2 IS SUBTRACTED FROM THE RESULT AND THE REMAINDER IS STORED IN 'ODD'. THE ADDRESS IN THE SR IS READ AGAIN, AND THE PROGRAM THEN WAITS FOR THE SWITCH OPTIONS TO BE SET, THE ADDRESS BEING SHOWN IN THE AC. WHEN 'CONTINUE' IS PRESSED, IF THE ADDRESS IS ODD IT IS INCREMENTED. 2 IS SUBTRACTED FROM THE RESULT AND THE REMAINDER IS STORED IN 'EVEN'. A SUBROUTINE THEN PUNCHES BLANK LEADER TAPE. THE LENGTH OF THE LEADER IS DETERMINED BY THE NUMBER STORED AT LL (16).

EACH NUMBER OF THE MINILOADER FORMAT IS BROUGHT INTO THE AC IN TURN AND PUNCHED AS TWO CHARACTERS BY A SUBROUTINE. THE ADDRESS OF EACH ORDER TO BE PUNCHED IS KEPT IN 'C' (167) AND IS TESTED BEFORE PUNCHING TAKES PLACE. WHEN THIS NUMBER REACHES 7762, BLANK TAPE AND THE FIRST ODD ADDRESS ARE PUNCHED. THE ODD NUMBERED ORDERS ARE THEN PUNCHED UNTIL 7763 IS REACHED, WHEN ONE BLANK CHARACTER IS PUNCHED FOLLOWED BY A TERMINATION DEPENDING ON THE CONTENT OF REGISTER 20.

IF REGISTER 20 = 0, EIGHT MORE BLANK CHARACTERS ARE PUNCHED, FOLLOWED BY NOP (7000), TO RESTORE THE MINILOADER TO A STARTING CONDITION BY LOADING NOP AT 7774. OTHERWISE, THE INSTRUCTION RECORDED IN 20 IS PUNCHED (TO BE LOADED AT 7764), FOLLOWED BY BLANK TRAILER.

5.1 EXECUTION TIME

THE SPEED OF THIS ROUTINE IS OUTPUT LIMITED.

6. SPECIAL FORMATS

6.1 EXTERNAL DATA

SEE 'MINILOADER' FOR A DESCRIPTION OF THE MINILOADER FORMAT.

6.2 THIS PROGRAM IS NORMALLY USED TO RECORD ONLY A BINARY LOADER, TO WHICH PROGRAM CONTROL IS TRANSFERRED WHEN IT IS LOADED FROM THE TAPE. IF IT IS USED WITH A TAPE COPYING PROGRAM, PROGRAMS IN BINARY FORMAT MAY BE ADDED TO THE MINILOADER FORMAT TAPE. THE MINILOADER CAN THEN BE USED AS THE STARTING SEQUENCE OF A BOOTSTRAP PAPER TAPE SYSTEM. THE TAPES FO 'BINARY LOADER 4B' ARE EXAMPLES OF THIS. HERE THE BINARY LOADER RE-LOADS THE MINILOADER IN ITS STARTING FORM. THIS BINARY LOADER CAN ALSO START THE PROGRAM IT LOADS, AND IT CAN SELECT FOR LOADING ANY OR ALL OF SEVERAL PROGRAMS OR OVERLAYS ON THE SAME TAPE (IN ACCORDANCE WITH ANY TEST WHICH CAN BE PROGRAMMED). IT SELECTS THE VERSION OF THE MINILOADER APPROPRIATE TO THE READER BEING USED. IT COULD CONTINUE BY LOADING ANOTHER PROGRAM, WITH SELECTED OVERLAYS, AND THEN START IT.

```
                /MINILOADER PUNCH          .          .          R.B.W.

                *16
0016  7540  LL,   -240     /LEADER LENGTH (16 INCHES)

0017  7402         HLT
0020  0000  J,     0       /FOR A SELF-STARTING TAPE, LOAD THE REQUIRED
0021  6026         PLS     /       JUMP INSTRUCTION HERE. IT MUST BE A
0022  6046         TLS     /       DIRECT JUMP TO PAGE 37 OR PAGE 0.
0023  7604         LAS
0024  7010         RAR
0025  7604         LAS
0026  7420         SNL
0027  7001         IAC
0030  1163         TAD M2  /COMPLEMENTS LINK
0031  3164         DCA ODD /FIRST ODD ADDRESS -2
0032  7604         LAS
0033  7402         HLT     /CHECK ADDRESS IN AC. SET SWITCH OPTIONS.
0034  7420         SNL
0035  7001         IAC
0036  1163         TAD M2
0037  3165         DCA EVEN/FIRST EVEN ADDRESS -2
0040  1137         TAD L
0041  3130         DCA H
0042  1016         TAD LL
0043  4121         JMS P
0044  1166         TAD ISZI
0045  4146         JMS PUNCH
0046  1165         TAD EVEN
0047  3167  T4,     DCA C
0050  1167         TAD C
0051  4146  T1,     JMS PUNCH
0052  1167  T3,     TAD C
0053  1170         TAD M7763
0054  7450         SNA
0055  5104         JMP END /       /WHEN C=7763
0056  7001         IAC
0057  7650         SNA CLA
0060  5100         JMP HALFWY      /WHEN C=7762
0061  1567         TAD I C
0062  7440         SZA
0063  5051         JMP T1  /REGISTER IS NOT 0
0064  7604         LAS     /REGISTER IS 0
0065  7004         RAL
0066  7710         SPA CLA
0067  5072         JMP T2  /SR1 IS 1
0070  1167         TAD C   /SR1 IS 0: HALT BEFORE THIS ZERO REGISTER IS
0071  7402         HLT     /       PUNCHED, WITH ITS ADDRESS IN AC
0072  7604  T2,     LAS
0073  7710         SPA CLA
0074  5051         JMP T1  /SR0 IS 1: TAPE WILL SET THIS REGISTER TO 0
0075  4126         JMS PUN /SR0 IS 0: THIS REGISTER WILL NOT BE ALTERED
0076  4126         JMS PUN
0077  5052         JMP T3
0100  1171  HALFWY, TAD M14
0101  4121         JMS P
0102  1164         TAD ODD
0103  5047         JMP T4
0104  4126  END,    JMS PUN /C > 7764
```

```
0105  1020          TAD J
0106  7450          SNA
0107  5112          JMP T5
0110  4146          JMS PUNCH
0111  5017          JMP J-1
0112  1172  T5,     TAD M10 /THIS SECTION RESTORES ANY OF THE THREE
0113  4121          JMS P   /       VERSIONS OF THE MINILOADER TO A
0114  1173          TAD N   /       STARTING CONDITION AND PUNCHES
0115  4146          JMS PUNCH /     BLANK TRAILER.
0116  1016          TAD LL
0117  4121          JMS P
0120  5017          JMP J-1

0121  0000  P,      0       /PUNCH BLANK TAPE
0122  3167          DCA C
0123  4126          JMS PUN
0124  5123          JMP .-1
0125  5521          JMP I P

0126  0000  PUN,    0       /PUNCH CONTENT OF AC, CLEAR AC, INCREMENT C.
0127  6021  HIGH,   PSF     /       WHEN C=0, INCREMENT PUN.
0130  5136  H,      JMP LOW
0131  6026          PLS
0132  7200          CLA
0133  1174          TAD J1
0134  3130          DCA H
0135  5142          JMP FINISH
0136  6041  LOW,    TSF
0137  5136  L,      JMP LOW
0140  6046          TLS
0141  7200          CLA
0142  2167  FINISH, ISZ C
0143  7410          SKP
0144  2126          ISZ PUN
0145  5526          JMP I PUN

0146  0000  PUNCH,  0       /PUNCH WORD IN AC AS TWO CHARACTERS AND
0147  3175          DCA TEM /       CLEAR AC
0150  1175          TAD TEM
0151  7120          STL
0152  7012          RTR
0153  0176          AND MASK
0154  7012          RTR
0155  7012          RTR
0156  4126          JMS PUN
0157  1175          TAD TEM
0160  0177          AND M
0161  4126          JMS PUN
0162  5546          JMP I PUNCH
```

```
0163  7776  M2,      -2
0164  0000  ODD,     0
0165  0000  EVEN,    0
0166  2376  ISZI,    2376     /"ISZ A" FOR THE MINILOADER
0167  0000  C,       0
0170  0015  M7763,   -7763
0171  7764  M14,     -14
0172  7770  M10,     -10
0173  7000  N,       NOP      /TO REPLACE "JMP B" AT 7774 TO BE COMPATIBLE
0174  5127  J1,      JMP HIGH/        WITH ALL VERSIONS OF THE MINILOADER.
0175  0000  TEM,     0
0176  3777  MASK,    3777
0177  0077  M,       77

            *17

C         0167
END       0104
EVEN      0165
FINISH    0142
H         0130
HALFWY    0100
HIGH      0127
ISZI      0166
J         0020
J1        0174
L         0137
LL        0016
LOW       0136
M         0177
MASK      0176
M10       0172
M14       0171
M2        0163
M7763     0170
N         0173
ODD       0164
P         0121
PUN       0126
PUNCH     0146
TEM       0175
T1        0051
T2        0072
T3        0052
T4        0047
T5        0112
```